

Description

`collect fromframe` puts values from a data frame into a collection. Values from numeric and string (`str#` and `strL`) variables can be collected. Tables can be constructed from these values in the collection.

`collect fromframe` does not affect the automatic result levels (automatic levels) for subsequent table layouts.

Quick start

Collect the values in a dataset created by collapsing the means of `v1` and `v2` by each level of `catvar`

```
collapse v1 v2, by(catvar)
collect fromframe v1 v2
```

Collect variables `v1` and `v2` from the current frame, and use variable `catvar` to create tags

```
collect fromframe v1 v2, dimensions(catvar)
```

Same as above, but collect variables from frame `fr1`

```
frame fr1: collect fromframe v1 v2, dimensions(catvar)
```

Menu

Statistics > Summaries, tables, and tests > Tables and collections > Collect from a frame

Syntax

```
collect fromframe varlist [if] [in] [, options]
```

<i>options</i>	Description
Main	
<code>name(<i>cname</i>)</code>	add values to collection <i>cname</i>
<code>dimensions(<i>dimvars</i>)</code>	select variables to tag collected values
<code>tags(<i>tags</i>)</code>	add custom tags

tags is a dimension and its corresponding level, or it is a space-separated list of dimensions and their corresponding levels:

```
dimname[levvalue] [dimname[levvalue] ...]
```

levvalue is treated as a single level of *dimname*, even if it contains a space and is not quoted.

Options

Main

`name(cname)` specifies the collection into which results will be saved, instead of the current collection.

`dimensions(dimvars)` specifies variables for constructing the tags attached to each collected value.

Variable and value labels attached to the variables in *dimvars* are applied to their corresponding dimensions in the collection.

If `dimensions()` is not specified, then all variables not in *varlist* are used to construct tags.

`tags(tags)` specifies additional tags to attach to the values being consumed. Each tag takes on the form *dimname*[*levvalue*]; to specify multiple tags, separate them with a space. The following dimension names are not allowed in `tags()`: `border_block`, `cell_type`, `header_cell_contents`, `program_class`, `result`, and `result_type`.

Remarks and examples

The first step in creating a table using `collect` is to collect the values that you wish to display in the table. Usually, this means using the `collect get` command or the `collect:` prefix to consume results (values) from one or more Stata commands. Use `collect fromframe` if the values you want to put in a table are in a data frame.

To collect values from the current data frame, type

```
. collect fromframe varlist
```

The variables specified in *varlist* contain the values we want to add to the collection. Values collected from each *var* in *varlist* will be tagged with `result[var]`. Without an *if* or *in* restriction, all observations will be added to the collection. If there are other variables in the dataset, they will be used to tag the individual values so they can be arranged into a table.

► Example 1

We have fictional data on grade-point averages (GPA) for college students in their first, second, third, and fourth years of college.

```
. use https://www.stata-press.com/data/r19/college
. describe
```

Contains data from https://www.stata-press.com/data/r19/college.dta

Observations: 12

Variables: 4 3 Jan 2024 12:05

Variable name	Storage type	Display format	Value label	Variable label
gpa	float	%9.0g		gpa for this year
hour	int	%9.0g		Total academic hours
year	int	%9.0g		1 = freshman, 2 = sophomore, 3 = junior, 4 = senior
number	int	%9.0g		number of students

Sorted by: year

We are interested in the mean and median GPA for each year; we collapse the dataset as follows:

```
. collapse (mean) mean=gpa (median) median=gpa, by(year)
```

Now the current frame contains variables named `mean` and `median` with mean and median GPAs. We collect the means and medians for each year by typing

```
. collect fromframe mean median
```

Every mean value in the dataset is tagged with `result[mean]` and `year[year]`, where *year* is a level of the year variable. Similarly, every median value in the dataset is collected and tagged with `result[median]` and `year[year]`.

With our values collected, we now specify a table layout by typing

```
. collect layout (year) (result)
```

Collection: default

Rows: year

Columns: result

Table 1: 4 x 2

	mean median	
1	2.9	3
2	3.066667	2.9
3	3.066667	3.3
4	3.15	3.15

See [\[TABLES\] collect layout](#) for information on specifying the table layout. For our purposes here, the important issue is that we have specified that the mean and median values, denoted by `result`, be placed on the columns.

Our process above was simple because we had only the three variables of interest in our frame: `mean`, `median`, and `year`. If there were other variables in the dataset, `collect fromframe` would also tag the collected values using their name and levels. If you do not want the other variables to be used in

the tags, use the `dimensions()` option to select the variables you want for the tags. For example, we might initially be interested in the average number of academic hours and include that result as well when collapsing our data:

```
. use https://www.stata-press.com/data/r19/college
. collapse (mean) meangpa=gpa meanhour=hour (median) medgpa=gpa, by(year)
```

If we issue the `collect fromframe` command, similarly to how we did above,

```
. collect fromframe meangpa medgpa
```

our collected values will also be tagged with the variable name `meanhour` and its levels. Instead, we can use the `dimensions()` option to restrict which variables are used to construct the tags.

```
. collect fromframe meangpa medgpa, dimensions(year)
```



Also see

[TABLES] **collect create** — Create a new collection

[TABLES] **collect get** — Collect results from a Stata command

[TABLES] **collect layout** — Specify table layout for the current collection

[TABLES] **collect set** — Set the current (active) collection

[TABLES] **collect unget** — Remove results from a collection

[R] **table intro** — Introduction to tables of frequencies, summaries, and command results

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).